```
g[x_, y_] := 1/(2 * π * σ^2) * ⅇ^(-(x^2+y^2)/(2*σ^2));
G[x_, y_] = FourierTransform[g[a, b], {a, b}, {x, y}];

animateGaussFrequency[σControl_] := Column[{
    Plot3D[{g[x, y] /. {σ -> σControl}, G[x, y] /. {σ -> σControl}}, {x, -3, 3}, {y, -3, 3},
     PlotStyle → {
       Directive[Specularity[■, 3], ■, Lighting →
         {{"Ambient", ■}, {"Directional", ■, ImageScaled[{0, 2, 2}]}, {"Directional", ■,
           ImageScaled[{2, 2, 2}]}, {"Directional", ■, ImageScaled[{2, 0, 2}]}}, Opacity[0.5]],
       Directive[Specularity[■, 3], ■, Lighting → {{"Ambient", ■}, {"Directional",
           ■, ImageScaled[{0, 2, 2}]}, {"Directional", ■, ImageScaled[{2, 2, 2}]},
          {"Directional", ■, ImageScaled[{2, 0, 2}]}}, Opacity[0.5]]
      },
     PlotRange → All,
     PlotLegends → {"spatial", "frequency"},
     AxesLabel → {"x,ω1", "y,ω2", "gσ,Gσ"},
     PlotLabel → "σ = " <> ToString[σControl],
     ImageSize → Large,
     AxesStyle → Medium,
     BaseStyle → {FontSize → 14}
    ],

    Plot[{g[x, 0] /. {σ -> σControl}, G[x, 0] /. {σ -> σControl}}, {x, -3, 3},
     PlotRange → All,
     PlotLegends → {"spatial", "frequency"},
     AxesLabel → {"x,ω1", "gσ(x,0),Gσ(ω1,0)"},
     ImageSize → Large,
     AxesStyle → Medium,
     BaseStyle → {FontSize → 14}
    ]
   }];

Manipulate[
 animateGaussFrequency[σ]
 , {{σ, 1.936}, 0.1, 3}]
```

σ

animateGaussFrequency[1.936]

```
(*Export[FileNameJoin[{NotebookDirectory[],"frames/sigma=00.png"}],
  Table[animateGaussFrequency[σ],{σ,0.136,3.036,0.1}],"VideoFrames",Antialiasing→True];*)
```

Simplified versions of the Gaussian function and the corresponding Fourier transform.

```
Simplify[g[x, y], σ ≥ 0]
```

$$\frac{e^{-\frac{x^2+y^2}{2\sigma^2}}}{2\pi\sigma^2}$$

```
Simplify[G[ω1, ω2], σ ≥ 0]
```

$$\frac{e^{-\frac{1}{2}\sigma^2\left(\omega_1^2+\omega_2^2\right)}}{2\pi}$$

Applying two Gaussian functions (here in the frequency domain) leads to...

```
2 π * (G[ω1, ω2] /. {σ → σs}) * (G[ω1, ω2] /. {σ → σx}) // Simplify
```

$$\frac{e^{-\frac{1}{2}\left(\sigma_s^2+\sigma_x^2\right)\left(\omega_1^2+\omega_2^2\right)}}{2\pi}$$

...one Gaussian with adjusted σ

```
G[ω1, ω2] /. {σ → Sqrt[σs^2 + σx^2]}
```

$$\frac{e^{-\frac{1}{2}\left(\sigma_s^2+\sigma_x^2\right)\left(\omega_1^2+\omega_2^2\right)}}{2\pi}$$

Test this in the spatial domain (with concrete numbers and on the special Dirac delta function)

```
Convolve[g[x, y] /. {σ → 10}, DiracDelta[x, y], {x, y}, {a, b}]
```

$$\frac{e^{\frac{1}{200}\left(-a^2-b^2\right)}}{200\pi}$$

```
Convolve[g[x, y] /. {σ → 2}, E^((-x^2 - y^2)/200)/(200 π), {x, y}, {a, b}]
```

$$\frac{e^{\frac{1}{208}\left(-a^2-b^2\right)}}{208\pi}$$

```
Convolve[g[x, y] /. {σ → Sqrt[10^2 + 2^2]}, DiracDelta[x, y], {x, y}, {a, b}]
```

$$\frac{e^{\frac{1}{208}\left(-a^2-b^2\right)}}{208\pi}$$

```
g[x, y] /. {σ → Sqrt[10^2 + 2^2]}
```

$$\frac{e^{\frac{1}{208}\left(-x^2-y^2\right)}}{208\pi}$$

And in the frequency domain

```
FourierTransform[DiracDelta[x, y], {x, y}, {ω1, ω2}]
```

$$\frac{1}{2\pi}$$

```
(1/(2 π) * 2 π * (G[ω1, ω2] /. {σ → 10})) * 2 π * (G[ω1, ω2] /. {σ → 2})
```

$$\frac{e^{-52\left(\omega_1^2+\omega_2^2\right)}}{2\pi}$$

```
InverseFourierTransform[E^(-52 (ω1^2 + ω2^2))/(2 π), {ω1, ω2}, {x, y}] // Simplify
```

$$\frac{e^{\frac{1}{208}\left(-x^2-y^2\right)}}{208\pi}$$

```
1/(2 π) * 2 π * G[ω1, ω2] /. {σ → Sqrt[10^2 + 2^2]}
```

$$\frac{e^{-52\left(\omega_1^2+\omega_2^2\right)}}{2\pi}$$

The used Gaussian is normalized

```
NIntegrate[g[x, y] /. {σ → 2}, {x, -100, 100}, {y, -100, 100}]
```

```
1.
```